
    Isabelle Elixir

Elixir clients for the Isabelle server.
The library speaks Isabelle's server protocol directly. See Chapter 4 in the Isabelle system manual for the specification.
Clients
IsabelleClientMini is the low-level building block. It is stateless, exposes
the TCP socket, and gives you explicit command/3, async_command/3, and
await_task/3 helpers.
IsabelleClient is the default client for scripts and notebooks. It keeps the
socket and current session_id in a struct, and awaits asynchronous Isabelle
tasks for the common session workflow.
IsabelleClientFull is a GenServer wrapper. It owns the socket, so callers
may safely share it across processes. Calls are serialized by design.
Tutorial Livebooks
The notebooks in livebook_examples/ are intended to be read and run in this
order:
	IsabelleClientMini.livemd introduces the wire-level building blocks and
explicit task handling.
	IsabelleClient.livemd shows the default stateful client for ordinary use.
	IsabelleClientFull.livemd shows the process-owning client and why it is
the right choice when multiple Elixir processes share one Isabelle
connection.

Together they serve as the tutorial for the library. They start local Isabelle
servers, run smoke tests, build and start a HOL session, check theories,
purge, stop, and clean up. The "Full" notebook additionally demonstrates
concurrency-safe access.
Example
Make sure Isabelle is available on PATH:
export PATH=/path/to/Isabelle2025-2/bin:$PATH

Then:
{:ok, [server]} = IsabelleClientMini.new_server("elixir", 0)

{:ok, client} =
  IsabelleClient.connect(server["password"],
    host: server["host"],
    port: server["port"]
  )

{:ok, client, _task} = IsabelleClient.start_session(client, %{"session" => "HOL"})

File.write!("/tmp/Example.thy", """
theory Example imports Main
begin

theorem "x = x"
  sledgehammer by simp

proposition "x = y"
  nitpick oops

end
""")

{:ok, task} =
  IsabelleClient.use_theories(client, %{
    "theories" => ["Example"],
    "master_dir" => "/tmp"
  })

IO.puts(IsabelleClientMini.extract_results(task))

{:ok, _client, _task} = IsabelleClient.stop_session(client)


  

    IsabelleClientMini

IsabelleClientMini is the bare-bones client. It exposes the TCP socket and
keeps no state. That is useful when you want a tiny protocol building block or
you want to own all lifecycle decisions yourself.
This notebook does the following:
	start and discover a local Isabelle server
	connect with the server password
	run help, echo, and raw command
	exercise cancel, session_build, session_start, use_theories,
purge_theories, session_stop, and shutdown
	extract messages from the finished theory task

Setup
When using the published package, replace the dependency with {:isabelle_elixir, "~> 0.2"}.
Mix.install([
  {:isabelle_elixir, path: Path.expand("..", __DIR__)}
])
You need to make sure that the isabelle executable is in the PATH.
System.find_executable("isabelle")
Start And Discover A Server
Passing port 0 asks Isabelle to choose a free local port. The returned map is
the connection descriptor: host, port, server name, and password.
server_name = "livebook_mini_#{System.unique_integer([:positive])}"
{:ok, [server]} = IsabelleClientMini.new_server(server_name, 0)
server
list_servers/0 goes through Isabelle's local server registry. This is handy
when a server was started outside of Elixir.
{:ok, servers} = IsabelleClientMini.list_servers()
Connect
Mini returns a raw socket. All later calls receive that socket explicitly.
{:ok, socket} =
  IsabelleClientMini.connect(server["password"], server["host"], server["port"])
socket
Synchronous Commands
help is a simple command. Isabelle may return the response as a short line or
as a byte-length-framed message; Mini hides that framing detail.
{:ok, commands} = IsabelleClientMini.help(socket)
commands
echo is useful because it round-trips arbitrary JSON values. The long string
forces Isabelle's long-message framing path.
{:ok, echoed} =
  IsabelleClientMini.echo(socket, %{
    "client" => "mini",
    "framed" => String.duplicate("x", 120)
  })
echoed
The raw command/3 escape hatch is still there for server commands that do not
deserve their own helper yet.
{:ok, %{"client" => "mini", "raw" => true}} =
  IsabelleClientMini.command(socket, "echo", %{"client" => "mini", "raw" => true})
Canceling an unknown task is a no-op on the Isabelle side, but it proves the
cancel command's empty OK body is handled correctly.
{:ok, nil} =
  IsabelleClientMini.cancel_task(socket, "00000000-0000-0000-0000-000000000000")
Build And Start A Session
The next commands are asynchronous. Isabelle first returns OK {"task": ...},
then streams NOTE messages, and finally returns FINISHED or FAILED.
{:ok, build_task} = IsabelleClientMini.build_session(socket, %{"session" => "HOL"})
{:ok, build_task} = IsabelleClientMini.await_task(socket, build_task, 120_000)

{build_task.status, build_task.result["ok"], length(build_task.notes)}
{:ok, start_task} = IsabelleClientMini.start_session(socket, %{"session" => "HOL"})
{:ok, start_task} = IsabelleClientMini.poll_status(socket, start_task, 120_000)

session_id = IsabelleClientMini.extract_session(start_task)
{start_task.status, session_id, Enum.map(start_task.notes, & &1["message"])}
Check A Theory
Isabelle reads theories from disk, so the notebook creates a temporary directory.
theory_dir =
  Path.join(System.tmp_dir!(), "isabelle_elixir_livebook_mini_#{System.unique_integer([:positive])}")
File.mkdir_p!(theory_dir)
theory_dir
We write a tiny .thy file to the temporary directory and ask the session to check it.
File.write!(Path.join(theory_dir, "Example.thy"), """
theory Example imports Main
begin

theorem "x = x"
  sledgehammer by simp

proposition "x = y"
  nitpick oops

end
""")
{:ok, use_task} =
  IsabelleClientMini.use_theories(socket, %{
    "session_id" => session_id,
    "theories" => ["Example"],
    "master_dir" => theory_dir
  })

{:ok, use_task} = IsabelleClientMini.await_task(socket, use_task, 120_000)
{use_task.status, use_task.result["ok"], use_task.result["errors"]}
extract_results/1 gives the user-facing messages from the checked node.
results = IsabelleClientMini.extract_results(use_task)
purge_theories is synchronous and returns the purged/retained node lists.
{:ok, purge_result} =
  IsabelleClientMini.purge_theories(socket, %{
    "session_id" => session_id,
    "theories" => ["Example"],
    "master_dir" => theory_dir
  })

{purge_result["purged"], purge_result["retained"]}
Stop And Shutdown
Stop the Isabelle session cleanly, then shut down the resident server through
the protocol.
{:ok, stop_task} = IsabelleClientMini.stop_session(socket, session_id)
{:ok, stop_task} = IsabelleClientMini.await_task(socket, stop_task, 120_000)

{stop_task.status, stop_task.result}
{:ok, nil} = IsabelleClientMini.shutdown_server(socket)
:ok = IsabelleClientMini.close(socket)
kill_server/1 is harmless after protocol shutdown and useful when rerunning
only part of the notebook.
IsabelleClientMini.kill_server(server_name)


  

    IsabelleClient

IsabelleClient is the ordinary user-facing client. It keeps the socket and
current session_id in a struct, which makes session workflows less noisy than
the stateless Mini API.
This notebook does the following:
	verify no-session errors
	run synchronous commands
	build and start HOL
	check a tiny theory without manually passing session_id
	purge, stop, and shut down cleanly

Setup
When using the published package, replace the dependency with {:isabelle_elixir, "~> 0.2"}.
Mix.install([
  {:isabelle_elixir, path: Path.expand("..", __DIR__)}
])
You need to make sure that the isabelle executable is in the PATH.
System.find_executable("isabelle")
Start And Connect
Server lifecycle still lives in IsabelleClientMini, because it is independent
from the client style used after connection.
server_name = "livebook_client_#{System.unique_integer([:positive])}"
{:ok, [server]} = IsabelleClientMini.new_server(server_name, 0)

{:ok, client} =
  IsabelleClient.connect(server["password"], host: server["host"], port: server["port"])

client
Guard Rails Before A Session Exists
The stateful client refuses session-dependent operations until a session has
actually been started.
{
  IsabelleClient.use_theories(client),
  IsabelleClient.purge_theories(client),
  IsabelleClient.stop_session(client)
}
Synchronous Commands
{:ok, commands} = IsabelleClient.help(client)
commands
{:ok, "some string"} = IsabelleClient.echo(client, "some string")
{:ok, %{"client" => "stateful", "raw" => true}} =
  IsabelleClient.command(client, "echo", %{"client" => "stateful", "raw" => true})
Build And Start A Session
The high-level client awaits async tasks for you in build_session/3,
start_session/3, use_theories/3, and stop_session/2.
{:ok, build_task} =
  IsabelleClient.build_session(client, %{"session" => "HOL"}, 120_000)

{build_task.status, build_task.result["ok"], length(build_task.notes)}
{:ok, client, start_task} =
  IsabelleClient.start_session(client, %{"session" => "HOL"}, 120_000)

{client.session_id, start_task.status, start_task.result}
Check A Theory
Isabelle reads theories from disk, so the notebook creates a temporary directory.
theory_dir =
  Path.join(System.tmp_dir!(), "isabelle_elixir_livebook_stateful_#{System.unique_integer([:positive])}")
File.mkdir_p!(theory_dir)
theory_dir
Unlike the "mini" client, this call does not need "session_id" in the argument map. The
client inserts its current session id.
File.write!(Path.join(theory_dir, "Example.thy"), """
theory Example imports Main
begin

theorem "x = x"
  sledgehammer by simp

proposition "x = y"
  nitpick oops

end
""")
{:ok, use_task} =
  IsabelleClient.use_theories(
    client,
    %{"theories" => ["Example"], "master_dir" => theory_dir},
    120_000
  )

{use_task.status, use_task.result["ok"], use_task.result["errors"]}
results = IsabelleClientMini.extract_results(use_task)
{:ok, purge_result} =
  IsabelleClient.purge_theories(client, %{
    "theories" => ["Example"],
    "master_dir" => theory_dir
  })

{purge_result["purged"], purge_result["retained"]}
Stop And Shutdown
Stopping the session returns an updated client with session_id: nil.
{:ok, client, stop_task} = IsabelleClient.stop_session(client, 120_000)
{client.session_id, stop_task.status, stop_task.result}
{:ok, nil} = IsabelleClient.shutdown_server(client)
:ok = IsabelleClient.close(client)
IsabelleClientMini.kill_server(server_name)


  

    IsabelleClientFull

IsabelleClientFull is the process-owning client. It wraps the stateful client
in a GenServer, so many Elixir processes can share one Isabelle connection
without racing on socket reads and writes.
The important distinction is socket ownership. Isabelle replies arrive on a
single TCP stream. In the other clients, if multiple processes call send/recv directly on the same
socket, or share the plain stateful client without coordination, a caller
can consume another caller's reply. Full prevents that by funneling all protocol
traffic through one owner process.
This notebook does the following:
	verify no-session errors
	run many concurrent unique long echo calls through one client process
	interleave concurrent help calls with those echoes
	build and start HOL
	check several distinct theories concurrently via use_theories
	purge theories, stop the session, shut down the server, and close the GenServer

Setup
When using the published package, replace the dependency with {:isabelle_elixir, "~> 0.2"}.
Mix.install([
  {:isabelle_elixir, path: Path.expand("..", __DIR__)}
])
You need to make sure that the isabelle executable is in the PATH.
System.find_executable("isabelle")
Start And Connect
The returned pid is the client handle. The process owns the socket.
server_name = "livebook_full_#{System.unique_integer([:positive])}"
{:ok, [server]} = IsabelleClientMini.new_server(server_name, 0)

{:ok, pid} =
  IsabelleClientFull.connect(
    password: server["password"],
    host: server["host"],
    port: server["port"]
  )

Process.alive?(pid)
Guard Rails Before A Session Exists
The Full client keeps session state inside the GenServer. Before a session is
started, session-dependent APIs fail locally with :no_session instead of
sending bad commands to Isabelle.
{
  IsabelleClientFull.use_theories(pid, %{}),
  IsabelleClientFull.purge_theories(pid, %{}),
  IsabelleClientFull.stop_session(pid)
}
Baseline Command
{:ok, commands} = IsabelleClientFull.help(pid)
commands
Concurrent Unique Replies
This is showcasing that the "full" client correctly serializes access to one shared Isabelle TCP connection. The dangerous thing to avoid is that, since Isabelle replies come back on a single socket stream, if multiple Elixir processes use the same raw socket directly (process A send command A, process B send command B), then process B might accidentally read the reply for command A.
In the code below, the echo tasks constitute the main test:
Each task sends a unique payload and expects that exact payload back. The payload includes a long string to force Isabelle's byte-length message framing path. The help tasks are interleaved to add some additional "noise".
The operations are shuffled, and every worker waits behind a small start
barrier before calling the client. That keeps task creation order from
dominating the GenServer mailbox order. Results are collected with receive,
so the notebook observes completion order instead of reconstructing the original
task list order with ordered Task.await/2 calls.
parent = self()
ref = make_ref()
release_ref = make_ref()

operations =
  (Enum.map(1..25, &{:echo, &1}) ++ Enum.map(1..5, &{:help, &1}))
  |> Enum.shuffle()

concurrent_tasks =
  for operation <- operations do
    Task.async(fn ->
      receive do
        {:go, ^release_ref} -> :ok
      end

      case operation do
        {:echo, n} ->
          payload = %{
            "client" => "full",
            "n" => n,
            "token" => "token-#{n}",
            "framed" => String.duplicate(Integer.to_string(rem(n, 10)), 160)
          }

          {:ok, ^payload} = IsabelleClientFull.echo(pid, payload)

        {:help, _n} ->
          {:ok, commands} = IsabelleClientFull.help(pid)

          for command <- ~w(cancel echo help purge_theories session_build session_start session_stop shutdown use_theories) do
            true = command in commands
          end
      end

      send(parent, {:full_concurrent_result, ref, operation})
    end)
  end

Enum.each(concurrent_tasks, &send(&1.pid, {:go, release_ref}))

concurrent_results =
  for _ <- 1..30 do
    receive do
      {:full_concurrent_result, ^ref, result} -> result
    after
      30_000 -> raise "timed out waiting for concurrent result"
    end
  end

expected =
  (Enum.map(1..25, &{:echo, &1}) ++ Enum.map(1..5, &{:help, &1}))

Enum.each(concurrent_tasks, &Task.await(&1, 1_000))
true = Enum.sort(concurrent_results) == Enum.sort(expected)

%{
  creation_order: operations,
  completion_order: concurrent_results,
  all_expected_results_seen?: Enum.sort(concurrent_results) == Enum.sort(expected)
}
Build And Start A Session
{:ok, build_task} =
  IsabelleClientFull.build_session(pid, %{"session" => "HOL"}, 120_000)

{build_task.status, build_task.result["ok"], length(build_task.notes)}
{:ok, start_task} =
  IsabelleClientFull.start_session(pid, %{"session" => "HOL"}, 120_000)

session_id = start_task.result["session_id"]
{start_task.status, session_id, Enum.map(start_task.notes, & &1["message"])}
Active Session Sanity Check
Now the GenServer has an active session_id in its internal client state. A
simple echo is enough to show the process and socket are still usable after
that state transition. The earlier concurrency batch already covers shared
caller safety.
{:ok, %{"phase" => "active_session"}} =
  IsabelleClientFull.echo(pid, %{"phase" => "active_session"})
Check A Theory
Isabelle reads theories from disk, so the notebook creates a temporary directory.
theory_dir =
  Path.join(System.tmp_dir!(), "isabelle_elixir_livebook_full_#{System.unique_integer([:positive])}")
File.mkdir_p!(theory_dir)
theory_dir
The "full" client stores session state inside the GenServer, so use_theories/3
receives no explicit "session_id".
This section also tests the interesting concurrency case for theory checking:
several callers all run use_theories, and each one internally awaits
Isabelle's async NOTE/FINISHED task stream. With a shared raw socket, these
callers could accidentally consume each other's task messages. With the "full" client, they
queue behind the GenServer owner and each receives only its own theory result.
theories = [
  {"Example1", "lemma \"x = x\"\n  by simp", "theorem ?x = ?x"},
  {"Example2", "lemma \"xs @ [] = xs\"\n  by simp", "theorem ?xs @ [] = ?xs"},
  {"Example3", "lemma \"(A \\<longrightarrow> A) \\<and> True\"\n  by simp",
   "theorem (?A \\<longrightarrow> ?A) \\<and> True"}
]

for {theory, body, _expected} <- theories do
  File.write!(Path.join(theory_dir, "#{theory}.thy"), """
  theory #{theory} imports Main

  begin

  #{body}

  end
  """)
end

File.ls!(theory_dir)
theory_ref = make_ref()
theory_release_ref = make_ref()

theory_operations =
  Enum.map(theories, fn {theory, _body, expected_result} ->
    {:use_theories, theory, expected_result}
  end)

theory_tasks =
  theory_operations
  |> Enum.shuffle()
  |> Enum.map(fn operation ->
    Task.async(fn ->
      receive do
        {:go, ^theory_release_ref} -> :ok
      end

      result =
        case operation do
          {:use_theories, theory, expected_result} ->
            {:ok, use_task} =
              IsabelleClientFull.use_theories(
                pid,
                %{"theories" => [theory], "master_dir" => theory_dir},
                120_000
              )

            true = use_task.status == :finished
            true = use_task.result["ok"]
            [%{"theory_name" => theory_name}] = use_task.result["nodes"]
            true = theory_name == "Draft.#{theory}"

            true =
              use_task
              |> IsabelleClientMini.extract_results()
              |> String.contains?(expected_result)

            {:use_theories, theory, expected_result, use_task}
        end

      send(parent, {:full_concurrent_result, theory_ref, result})
    end)
  end)

Enum.each(theory_tasks, &send(&1.pid, {:go, theory_release_ref}))

theory_results =
  for _ <- 1..length(theory_tasks) do
    receive do
      {:full_concurrent_result, ^theory_ref, result} -> result
    after
      120_000 -> raise "timed out waiting for theory concurrency result"
    end
  end

Enum.each(theory_tasks, &Task.await(&1, 1_000))

checked_theories =
  theory_results
  |> Enum.map(fn {:use_theories, theory, _expected_result, _task} -> theory end)
  |> Enum.sort()

%{
  completion_order:
    Enum.map(theory_results, fn
      {:use_theories, theory, _expected_result, _task} -> theory
    end),
  all_expected_theories_seen?: checked_theories == ~w(Example1 Example2 Example3),
  theory_messages:
    Enum.map(theory_results, fn {:use_theories, theory, expected_result, task} ->
      %{
        theory: theory,
        expected_result: expected_result,
        messages: IsabelleClientMini.extract_results(task)
      }
    end)
}
{:ok, purge_result} =
  IsabelleClientFull.purge_theories(pid, %{
    "theories" => ~w(Example1 Example2 Example3),
    "master_dir" => theory_dir
  })

{purge_result["purged"], purge_result["retained"]}
Stop And Shutdown
{:ok, stop_task} = IsabelleClientFull.stop_session(pid, 120_000)
{stop_task.status, stop_task.result}
{:ok, nil} = IsabelleClientFull.shutdown_server(pid)
:ok = IsabelleClientFull.close(pid)
kill_server/1 is harmless after protocol shutdown and useful when rerunning
only part of the notebook.
IsabelleClientMini.kill_server(server_name)

Stateful, single-process Isabelle client.
This is the ergonomic client for scripts and LiveBooks. It keeps the socket
and current session_id in a struct, but it is not meant to be shared by
multiple concurrent processes. Use IsabelleClientFull for that.

    Functions
  


    
      
        build_session(isabelle_client, args, timeout \\ :infinity)

      


        Builds an Isabelle session image and waits for the task result.



    


    
      
        close(isabelle_client)

      


        Closes the client's socket.



    


    
      
        command(isabelle_client, name, arg \\ nil)

      


        Runs a synchronous Isabelle command.



    


    
      
        connect(password, opts \\ [])

      


        Connects to an Isabelle server and returns a stateful client struct.



    


    
      
        echo(client, value)

      


        Round-trips a JSON value through Isabelle's echo command.



    


    
      
        help(client)

      


        Returns the server command names supported by Isabelle.



    


    
      
        purge_theories(isabelle_client)

      


        Purges theories from the active session.



    


    
      
        purge_theories(isabelle_client, args)

      


    


    
      
        shutdown_server(client)

      


        Asks the Isabelle server process to shut down.



    


    
      
        start_session(client, args, timeout \\ :infinity)

      


        Starts an Isabelle session, stores its session_id, and returns the updated client.



    


    
      
        stop_session(isabelle_client)

      


        Stops the active Isabelle session and clears session_id.



    


    
      
        stop_session(client, timeout \\ :infinity)

      


    


    
      
        use_theories(client, args \\ nil, timeout \\ :infinity)

      


        Checks theories in the active session and waits for the task result.

          @type t() :: %IsabelleClient{session_id: String.t() | nil, socket: port()}

Builds an Isabelle session image and waits for the task result.

Closes the client's socket.

Runs a synchronous Isabelle command.

Connects to an Isabelle server and returns a stateful client struct.

Round-trips a JSON value through Isabelle's echo command.

Returns the server command names supported by Isabelle.

Asks the Isabelle server process to shut down.

  



    

  
    
      
    
    
      start_session(client, args, timeout \\ :infinity)



        
          
        

    

  


  

Starts an Isabelle session, stores its session_id, and returns the updated client.

Stops the active Isabelle session and clears session_id.

Checks theories in the active session and waits for the task result.

Local Isabelle server lifecycle helpers.

        kill(name)

      


        Force-kills a local resident Isabelle server by name.



    


    
      
        list()

      


        Lists local resident Isabelle servers.



    


    
      
        parse_info(data)

      


        Parses isabelle server output into server descriptor maps.



    


    
      
        start(name \\ "elixir", port \\ 9999)

      


        Starts a local resident Isabelle server.

Parses isabelle server output into server descriptor maps.

  



    

    

  
    
      
    
    
      start(name \\ "elixir", port \\ 9999)



        
          
        

    

  


  

Starts a local resident Isabelle server.

Result of an Isabelle asynchronous server task.

        Creates a running task struct from an Isabelle task id.



    





      


      
        Types

        


  
    
      
    
    
      t()



        
          
        

    

  


  

      

          @type t() :: %IsabelleClient.Task{
  id: String.t(),
  notes: [map()],
  result: map() | nil,
  status: :running | :finished | :failed
}

Creates a running task struct from an Isabelle task id.

GenServer-backed Isabelle client.
The process owns the TCP socket, so callers may safely use the same client
from multiple processes. Operations are serialized deliberately: Isabelle's
server can run asynchronous tasks, but a single TCP stream is still easiest
to reason about when one process owns reads and writes.

      


      
        Summary


  
    Functions
  


    
      
        build_session(server, args, timeout \\ :infinity)

      


        Builds an Isabelle session image and waits for the task result.



    


    
      
        child_spec(init_arg)

      


        Returns a specification to start this module under a supervisor.



    


    
      
        close(server)

      


        Stops the client process and closes its socket.



    


    
      
        command(server, name, arg \\ nil, timeout \\ :infinity)

      


        Runs a synchronous Isabelle command through the client process.



    


    
      
        connect(opts)

      


        Alias for start_link/1.



    


    
      
        echo(server, value)

      


        Round-trips a JSON value through Isabelle's echo command.



    


    
      
        help(server)

      


        Returns the server command names supported by Isabelle.



    


    
      
        purge_theories(server, args, timeout \\ :infinity)

      


        Purges theories from the active session.



    


    
      
        shutdown_server(server)

      


        Asks the Isabelle server process to shut down.



    


    
      
        start_link(opts)

      


        Starts a GenServer-backed Isabelle client.



    


    
      
        start_session(server, args, timeout \\ :infinity)

      


        Starts an Isabelle session and stores its session_id in the client process.



    


    
      
        stop_session(server, timeout \\ :infinity)

      


        Stops the active Isabelle session.



    


    
      
        use_theories(server, args, timeout \\ :infinity)

      


        Checks theories in the active session and waits for the task result.



    





      


      
        Functions

        


    

  
    
      
    
    
      build_session(server, args, timeout \\ :infinity)



        
          
        

    

  


  

Builds an Isabelle session image and waits for the task result.

  



  
    
      
    
    
      child_spec(init_arg)



        
          
        

    

  


  

Returns a specification to start this module under a supervisor.
See Supervisor.

  



  
    
      
    
    
      close(server)



        
          
        

    

  


  

Stops the client process and closes its socket.

  



    

    

  
    
      
    
    
      command(server, name, arg \\ nil, timeout \\ :infinity)



        
          
        

    

  


  

Runs a synchronous Isabelle command through the client process.

  



  
    
      
    
    
      connect(opts)



        
          
        

    

  


  

Alias for start_link/1.

  



  
    
      
    
    
      echo(server, value)



        
          
        

    

  


  

Round-trips a JSON value through Isabelle's echo command.

  



  
    
      
    
    
      help(server)



        
          
        

    

  


  

Returns the server command names supported by Isabelle.

  



    

  
    
      
    
    
      purge_theories(server, args, timeout \\ :infinity)



        
          
        

    

  


  

Purges theories from the active session.

  



  
    
      
    
    
      shutdown_server(server)



        
          
        

    

  


  

Asks the Isabelle server process to shut down.

  



  
    
      
    
    
      start_link(opts)



        
          
        

    

  


  

Starts a GenServer-backed Isabelle client.

  



    

  
    
      
    
    
      start_session(server, args, timeout \\ :infinity)



        
          
        

    

  


  

Starts an Isabelle session and stores its session_id in the client process.

  



    

  
    
      
    
    
      stop_session(server, timeout \\ :infinity)



        
          
        

    

  


  

Stops the active Isabelle session.

  



    

  
    
      
    
    
      use_theories(server, args, timeout \\ :infinity)



        
          
        

    

  


  

Checks theories in the active session and waits for the task result.

  


        

      


  

    
IsabelleClientMini 
    



      
Minimal stateless client for the Isabelle server.
The module intentionally exposes the TCP socket and keeps no process state.
It does, however, implement Isabelle's real line-message framing: replies may
be either a single line (OK ...) or a byte length line followed by exactly
that many bytes.

      


      
        Summary


  
    Functions
  


    
      
        async_command(socket, name, arg, timeout \\ 30000)

      


        Starts an asynchronous Isabelle command and returns an %IsabelleClient.Task{}.



    


    
      
        await_task(socket, task_or_id, timeout \\ :infinity)

      


        Waits for an asynchronous Isabelle task to finish or fail.



    


    
      
        build_session(socket, args)

      


        Starts an Isabelle session_build task.



    


    
      
        cancel_task(socket, task_id)

      


        Requests cancellation of an Isabelle asynchronous task.



    


    
      
        close(socket)

      


        Closes a raw Isabelle TCP socket.



    


    
      
        command(socket, name, arg \\ nil, timeout \\ 30000)

      


        Runs a synchronous Isabelle server command and returns its OK body.



    


    
      
        connect(password, host \\ "127.0.0.1", port \\ 9999, timeout \\ 30000)

      


        Connects to an Isabelle server and authenticates with its password.



    


    
      
        echo(socket, value)

      


        Round-trips a JSON value through Isabelle's echo command.



    


    
      
        extract_results(arg1)

      


        Extracts user-facing theory messages from a finished use_theories result.



    


    
      
        extract_session(arg1)

      


        Extracts the session_id from a finished session-start result.



    


    
      
        help(socket)

      


        Returns the server command names supported by Isabelle.



    


    
      
        kill_server(name)

      


        Force-kills a local resident Isabelle server by name.



    


    
      
        list_servers()

      


        Lists local resident Isabelle servers known to Isabelle.



    


    
      
        new_server(name \\ "elixir", port \\ 9999)

      


        Starts a local resident Isabelle server via isabelle server.



    


    
      
        poll_status(socket, task_or_id, timeout \\ :infinity)

      


        Alias for await_task/3, kept for the original Mini workflow.



    


    
      
        purge_theories(socket, args)

      


        Runs Isabelle purge_theories for an active session.



    


    
      
        recv(socket, timeout \\ 30000)

      


        Receives one framed Isabelle server response.



    


    
      
        shutdown_server(socket)

      


        Asks the Isabelle server process to shut down.



    


    
      
        start_session(socket, args)

      


        Starts an Isabelle session_start task.



    


    
      
        stop_session(socket, session_id)

      


        Starts an Isabelle session_stop task for a session id.



    


    
      
        use_theories(socket, args)

      


        Starts an Isabelle use_theories task.



    





      


      
        Functions

        


    

  
    
      
    
    
      async_command(socket, name, arg, timeout \\ 30000)



        
          
        

    

  


  

Starts an asynchronous Isabelle command and returns an %IsabelleClient.Task{}.

  



    

  
    
      
    
    
      await_task(socket, task_or_id, timeout \\ :infinity)



        
          
        

    

  


  

Waits for an asynchronous Isabelle task to finish or fail.

  



  
    
      
    
    
      build_session(socket, args)



        
          
        

    

  


  

Starts an Isabelle session_build task.

  



  
    
      
    
    
      cancel_task(socket, task_id)



        
          
        

    

  


  

Requests cancellation of an Isabelle asynchronous task.

  



  
    
      
    
    
      close(socket)



        
          
        

    

  


  

Closes a raw Isabelle TCP socket.

  



    

    

  
    
      
    
    
      command(socket, name, arg \\ nil, timeout \\ 30000)



        
          
        

    

  


  

Runs a synchronous Isabelle server command and returns its OK body.

  



    

    

    

  
    
      
    
    
      connect(password, host \\ "127.0.0.1", port \\ 9999, timeout \\ 30000)



        
          
        

    

  


  

Connects to an Isabelle server and authenticates with its password.

  



  
    
      
    
    
      echo(socket, value)



        
          
        

    

  


  

Round-trips a JSON value through Isabelle's echo command.

  



  
    
      
    
    
      extract_results(arg1)



        
          
        

    

  


  

Extracts user-facing theory messages from a finished use_theories result.

  



  
    
      
    
    
      extract_session(arg1)



        
          
        

    

  


  

Extracts the session_id from a finished session-start result.

  



  
    
      
    
    
      help(socket)



        
          
        

    

  


  

Returns the server command names supported by Isabelle.

  



  
    
      
    
    
      kill_server(name)



        
          
        

    

  


  

Force-kills a local resident Isabelle server by name.

  



  
    
      
    
    
      list_servers()



        
          
        

    

  


  

Lists local resident Isabelle servers known to Isabelle.

  



    

    

  
    
      
    
    
      new_server(name \\ "elixir", port \\ 9999)



        
          
        

    

  


  

Starts a local resident Isabelle server via isabelle server.

  



    

  
    
      
    
    
      poll_status(socket, task_or_id, timeout \\ :infinity)



        
          
        

    

  


  

Alias for await_task/3, kept for the original Mini workflow.

  



  
    
      
    
    
      purge_theories(socket, args)



        
          
        

    

  


  

Runs Isabelle purge_theories for an active session.

  



    

  
    
      
    
    
      recv(socket, timeout \\ 30000)



        
          
        

    

  


  

Receives one framed Isabelle server response.

  



  
    
      
    
    
      shutdown_server(socket)



        
          
        

    

  


  

Asks the Isabelle server process to shut down.

  



  
    
      
    
    
      start_session(socket, args)



        
          
        

    

  


  

Starts an Isabelle session_start task.

  



  
    
      
    
    
      stop_session(socket, session_id)



        
          
        

    

  


  

Starts an Isabelle session_stop task for a session id.

  



  
    
      
    
    
      use_theories(socket, args)



        
          
        

    

  


  

Starts an Isabelle use_theories task.

